Rules for integrands involving logarithms

1. $\int u \frac{\text{Log}[1-F[x]]\, F'[x]}{F[x]} \, dx$

1: $\int \frac{\text{Log}[1-F[x]]\, F'[x]}{F[x]} \, dx$

- Basis: $\partial_x \text{PolyLog}[2,\, x] == \frac{\text{PolyLog}[1,x]}{x} == -\frac{\text{Log}[1-x]}{x}$
- Rule:

$$\int \frac{\text{Log}[1-F[x]]\, F'[x]}{F[x]} \, dx \longrightarrow -\text{PolyLog}[2,\, F[x]]$$

- Program code:

```
Int[u_*Log[v_],x_Symbol] :=
  With[{w=DerivativeDivides[v,u*(1-v),x]},
  w*PolyLog[2,1-v] /;
 Not[FalseQ[w]]]
```

2: $\int (a+b\,\text{Log}[u]) \frac{\text{Log}[1-F[x]]\, F'[x]}{F[x]} \, dx$ when u is free of inverse functions

- Derivation: Integration by parts
- Basis: $\frac{\text{Log}[1-x]}{x} == -\partial_x \text{PolyLog}[2,\, x]$
- Rule: If u is free of inverse functions, then

$$\int (a+b\,\text{Log}[u]) \frac{\text{Log}[1-F[x]]\, F'[x]}{F[x]} \, dx \longrightarrow -(a+b\,\text{Log}[u])\, \text{PolyLog}[2,\, F[x]] + b \int \frac{\text{PolyLog}[2,\, F[x]]\, \partial_x u}{u} \, dx$$

- Program code:

```
Int[(a_.+b_.*Log[u_])*Log[v_]*w_,x_Symbol] :=
  With[{z=DerivativeDivides[v,w*(1-v),x]},
  z*(a+b*Log[u])*PolyLog[2,1-v] -
  b*Int[SimplifyIntegrand[z*PolyLog[2,1-v]*D[u,x]/u,x],x] /;
 Not[FalseQ[z]]] /;
FreeQ[{a,b},x] && InverseFunctionFreeQ[u,x]
```

2. $\int u\,(a+b\,\text{Log}[c\,\text{Log}[d\,x^n]^p])\,dx$

1: $\int \text{Log}[c\,\text{Log}[d\,x^n]^p]\,dx$

Derivation: Integration by parts

Basis: $\partial_x \text{Log}[c\,\text{Log}[d\,x^n]^p] == \frac{n\,p}{x\,\text{Log}[d\,x^n]}$

Rule:

$$\int \text{Log}[c\,\text{Log}[d\,x^n]^p]\,dx \longrightarrow x\,\text{Log}[c\,\text{Log}[d\,x^n]^p] - n\,p\int \frac{1}{\text{Log}[d\,x^n]}\,dx$$

Program code:

```
Int[Log[c_.*Log[d_.*x_^n_.]^p_.],x_Symbol] :=
  x*Log[c*Log[d*x^n]^p] - n*p*Int[1/Log[d*x^n],x] /;
FreeQ[{c,d,n,p},x]
```

2. $\int (e\,x)^m\,(a+b\,\text{Log}[c\,\text{Log}[d\,x^n]^p])\,dx$

1: $\int \frac{a+b\,\text{Log}[c\,\text{Log}[d\,x^n]^p]}{x}\,dx$

Derivation: Integration by parts

- **Basis:** $\frac{1}{x} == \partial_x \frac{\text{Log}[d\,x^n]}{n}$
- **Basis:** $\partial_x (a+b\,\text{Log}[c\,\text{Log}[d\,x^n]^p]) == \frac{b\,n\,p}{x\,\text{Log}[d\,x^n]}$

Rule:

$$\int \frac{a+b\,\text{Log}[c\,\text{Log}[d\,x^n]^p]}{x}\,dx \longrightarrow \frac{\text{Log}[d\,x^n]\,(a+b\,\text{Log}[c\,\text{Log}[d\,x^n]^p])}{n} - b\,p\int \frac{1}{x}\,dx \longrightarrow \frac{\text{Log}[d\,x^n]\,(a+b\,\text{Log}[c\,\text{Log}[d\,x^n]^p])}{n} - b\,p\,\text{Log}[x]$$

Program code:

```
Int[(a_.+b_.*Log[c_.*Log[d_.*x_^n_.]^p_.])/x_,x_Symbol] :=
  Log[d*x^n]*(a+b*Log[c*Log[d*x^n]^p])/n - b*p*Log[x] /;
FreeQ[{a,b,c,d,n,p},x]
```

2: $\int (e\,x)^m\,(a+b\,\mathrm{Log}[c\,\mathrm{Log}[d\,x^n]^p])\,dx$ when $m\neq -1$

Derivation: Integration by parts

- **Basis:** $\partial_x\,(a+b\,\mathrm{Log}[c\,\mathrm{Log}[d\,x^n]^p]) == \frac{b\,n\,p}{x\,\mathrm{Log}[d\,x^n]}$

Rule: If $m\neq -1$**, then**

$$\int (e\,x)^m\,(a+b\,\mathrm{Log}[c\,\mathrm{Log}[d\,x^n]^p])\,dx \longrightarrow \frac{(e\,x)^{m+1}\,(a+b\,\mathrm{Log}[c\,\mathrm{Log}[d\,x^n]^p])}{e\,(m+1)}-\frac{b\,n\,p}{m+1}\int\frac{(e\,x)^m}{\mathrm{Log}[d\,x^n]}\,dx$$

Program code:

```
Int[(e_.*x_)^m_.*(a_.+b_.*Log[c_.*Log[d_.*x_^n_.]^p_.]),x_Symbol] :=
  (e*x)^(m+1)*(a+b*Log[c*Log[d*x^n]^p])/(e*(m+1)) - b*n*p/(m+1)*Int[(e*x)^m/Log[d*x^n],x] /;
FreeQ[{a,b,c,d,e,m,n,p},x] && NeQ[m,-1]
```

3. $\int u\,(a+b\,\mathrm{Log}[c\,RF_x^p])^n\,dx$ when $n\in\mathbb{Z}^+$

1: $\int (a+b\,\mathrm{Log}[c\,RF_x^p])^n\,dx$ when $n\in\mathbb{Z}^+$

Derivation: Integration by parts

- **Basis:** $\partial_x\,(a+b\,\mathrm{Log}[c\,RF_x^p])^n == \frac{b\,n\,p\,(a+b\,\mathrm{Log}[c\,RF_x^p])^{n-1}\,\partial_x RF_x}{RF_x}$

Rule: If $n\in\mathbb{Z}^+$**, then**

$$\int (a+b\,\mathrm{Log}[c\,RF_x^p])^n\,dx \longrightarrow x\,(a+b\,\mathrm{Log}[c\,RF_x^p])^n-b\,n\,p\int\frac{x\,(a+b\,\mathrm{Log}[c\,RF_x^p])^{n-1}\,\partial_x RF_x}{RF_x}\,dx$$

Program code:

```
Int[(a_.+b_.*Log[c_.*RFx_^p_.])^n_.,x_Symbol] :=
  x*(a+b*Log[c*RFx^p])^n -
  b*n*p*Int[SimplifyIntegrand[x*(a+b*Log[c*RFx^p])^(n-1)*D[RFx,x]/RFx,x],x] /;
FreeQ[{a,b,c,p},x] && RationalFunctionQ[RFx,x] && IGtQ[n,0]
```

2. $\int (d+e\,x)^m\,(a+b\,\mathrm{Log}[c\,RF_x^p])^n\,dx$ when $n\in\mathbb{Z}^+\land(n==1\lor m\in\mathbb{Z})$

1: $\int \frac{(a+b\,\text{Log}[c\,RF_x{}^p])^n}{d+e\,x}\,dx$ when $n \in \mathbb{Z}^+$?? ?? n>1?

Derivation: Integration by parts

- **Basis:** $\frac{1}{d+e\,x} == \partial_x \frac{\text{Log}[d+e\,x]}{e}$
- **Basis:** $\partial_x\,(a+b\,\text{Log}[c\,RF_x{}^p])^n == \frac{b\,n\,p\,(a+b\,\text{Log}[c\,RF_x{}^p])^{n-1}\,\partial_x RF_x}{RF_x}$

Rule: If $n \in \mathbb{Z}^+$**, then**

$$\int \frac{(a+b\,\text{Log}[c\,RF_x{}^p])^n}{d+e\,x}\,dx \longrightarrow \frac{\text{Log}[d+e\,x]\,(a+b\,\text{Log}[c\,RF_x{}^p])^n}{e} - \frac{b\,n\,p}{e}\int \frac{\text{Log}[d+e\,x]\,(a+b\,\text{Log}[c\,RF_x{}^p])^{n-1}\,\partial_x RF_x}{RF_x}\,dx$$

Program code:

```
Int[(a_.+b_.*Log[c_.*RFx_^p_.])^n_./(d_.+e_.*x_),x_Symbol] :=
  Log[d+e*x]*(a+b*Log[c*RFx^p])^n/e -
  b*n*p/e*Int[Log[d+e*x]*(a+b*Log[c*RFx^p])^(n-1)*D[RFx,x]/RFx,x] /;
FreeQ[{a,b,c,d,e,p},x] && RationalFunctionQ[RFx,x] && IGtQ[n,0]
```

2: $\int (d+e\,x)^m\,(a+b\,\text{Log}[c\,RF_x{}^p])^n\,dx$ when $n \in \mathbb{Z}^+ \bigwedge (n == 1 \bigvee m \in \mathbb{Z}) \bigwedge m \neq -1$

Derivation: Integration by parts

- **Basis:** $(d+e\,x)^m == \partial_x \frac{(d+e\,x)^{m+1}}{e\,(m+1)}$
- **Basis:** $\partial_x\,(a+b\,\text{Log}[c\,RF_x{}^p])^n == \frac{b\,n\,p\,(a+b\,\text{Log}[c\,RF_x{}^p])^{n-1}\,\partial_x RF_x}{RF_x}$

Rule: If $n \in \mathbb{Z}^+ \bigwedge (n == 1 \vee m \in \mathbb{Z}) \bigwedge m \neq -1$**, then**

$$\int (d+e\,x)^m\,(a+b\,\text{Log}[c\,RF_x{}^p])^n\,dx \longrightarrow \frac{(d+e\,x)^{m+1}\,(a+b\,\text{Log}[c\,RF_x{}^p])^n}{e\,(m+1)} - \frac{b\,n\,p}{e\,(m+1)}\int \frac{(d+e\,x)^{m+1}\,(a+b\,\text{Log}[c\,RF_x{}^p])^{n-1}\,\partial_x RF_x}{RF_x}\,dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_.*(a_.+b_.*Log[c_.*RFx_^p_.])^n_.,x_Symbol] :=
  (d+e*x)^(m+1)*(a+b*Log[c*RFx^p])^n/(e*(m+1)) -
  b*n*p/(e*(m+1))*Int[SimplifyIntegrand[(d+e*x)^(m+1)*(a+b*Log[c*RFx^p])^(n-1)*D[RFx,x]/RFx,x],x] /;
FreeQ[{a,b,c,d,e,m,p},x] && RationalFunctionQ[RFx,x] && IGtQ[n,0] && (EqQ[n,1] || IntegerQ[m]) && NeQ[m,-1]
```

3: $\int \frac{\text{Log}[c\, RF_x{}^n]}{d+e\,x^2}\, dx$

Derivation: Integration by parts

Rule: Let $u = \int \frac{1}{d+e\,x^2}\, dx$**, then**

$$\int \frac{\text{Log}[c\, RF_x{}^n]}{d+e\,x^2}\, dx \longrightarrow u\, \text{Log}[c\, RF_x{}^n] - n \int \frac{u\, \partial_x RF_x}{RF_x}\, dx$$

Program code:

```
Int[Log[c_.*RFx_^n_.]/(d_+e_.*x_^2),x_Symbol] :=
  With[{u=IntHide[1/(d+e*x^2),x]},
  u*Log[c*RFx^n] - n*Int[SimplifyIntegrand[u*D[RFx,x]/RFx,x],x]] /;
FreeQ[{c,d,e,n},x] && RationalFunctionQ[RFx,x] && Not[PolynomialQ[RFx,x]]
```

4: $\int \frac{\text{Log}[c\, P_x{}^n]}{Q_x}\, dx$ when $\text{QuadraticQ}[Q_x] \bigwedge \partial_x \frac{P_x}{Q_x} == 0$

Derivation: Integration by parts

Rule: If $\text{QuadraticQ}[Q_x] \bigwedge \partial_x \frac{P_x}{Q_x} == 0$**, let** $u = \int \frac{1}{Q_x}\, dx$**, then**

$$\int \frac{\text{Log}[c\, P_x{}^n]}{Q_x}\, dx \longrightarrow u\, \text{Log}[c\, P_x{}^n] - n \int \frac{u\, \partial_x P_x}{P_x}\, dx$$

Program code:

```
Int[Log[c_.*Px_^n_.]/Qx_,x_Symbol] :=
  With[{u=IntHide[1/Qx,x]},
  u*Log[c*Px^n] - n*Int[SimplifyIntegrand[u*D[Px,x]/Px,x],x]] /;
FreeQ[{c,n},x] && QuadraticQ[{Qx,Px},x] && EqQ[D[Px/Qx,x],0]
```

5: $\int RG_x\,(a+b\,\text{Log}[c\,RF_x{}^p])^n\,dx$ when $n\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $n\in\mathbb{Z}^+$, then

$$\int RG_x\,(a+b\,\text{Log}[c\,RF_x{}^p])^n\,dx\;\longrightarrow\;\int(a+b\,\text{Log}[c\,RF_x{}^p])^n\,\text{ExpandIntegrand}[RG_x,\,x]\,dx$$

Program code:

```
Int[RGx_*(a_.+b_.*Log[c_.*RFx_^p_.])^n_.,x_Symbol] :=
  With[{u=ExpandIntegrand[(a+b*Log[c*RFx^p])^n,RGx,x]},
  Int[u,x] /;
 SumQ[u]] /;
FreeQ[{a,b,c,p},x] && RationalFunctionQ[RFx,x] && RationalFunctionQ[RGx,x] && IGtQ[n,0]
```

```
Int[RGx_*(a_.+b_.*Log[c_.*RFx_^p_.])^n_.,x_Symbol] :=
  With[{u=ExpandIntegrand[RGx*(a+b*Log[c*RFx^p])^n,x]},
  Int[u,x] /;
 SumQ[u]] /;
FreeQ[{a,b,c,p},x] && RationalFunctionQ[RFx,x] && RationalFunctionQ[RGx,x] && IGtQ[n,0]
```

4: $\int RF_x\left(a+b\,\text{Log}\left[F\left[(c+d\,x)^{1/n},\,x\right]\right]\right)dx$ when $n\in\mathbb{Z}$

Derivation: Integration by substitution

Basis: If $n\in\mathbb{Z}$, then $F\left[(c+d\,x)^{1/n},\,x\right]=\frac{n}{d}\,\text{Subst}\left[x^{n-1}\,F\left[x,\,-\frac{c}{d}+\frac{x^n}{d}\right],\,x,\,(c+d\,x)^{1/n}\right]\partial_x\,(c+d\,x)^{1/n}$

Rule: If $n\in\mathbb{Z}$, then

$$\int RF_x\left(a+b\,\text{Log}\left[F\left[(c+d\,x)^{1/n},\,x\right]\right]\right)dx\;\longrightarrow\;\frac{n}{d}\,\text{Subst}\left[\int x^{n-1}\,\text{Subst}\left[RF_x,\,x,\,-\frac{c}{d}+\frac{x^n}{d}\right]\left(a+b\,F\left[x,\,-\frac{c}{d}+\frac{x^n}{d}\right]\right)dx,\,x,\,(c+d\,x)^{1/n}\right]$$

Program code:

```
Int[RFx_*(a_.+b_.*Log[u_]),x_Symbol] :=
  With[{lst=SubstForFractionalPowerOfLinear[RFx*(a+b*Log[u]),x]},
  lst[[2]]*lst[[4]]*Subst[Int[lst[[1]],x],x,lst[[3]]^(1/lst[[2]])] /;
 Not[FalseQ[lst]]] /;
FreeQ[{a,b},x] && RationalFunctionQ[RFx,x]
```

5. $\int (f+g\,x)^m \operatorname{Log}\left[d+e\,\left(F^{c\,(a+b\,x)}\right)^n\right] dx$

1: $\int (f+g\,x)^m \operatorname{Log}\left[1+e\,\left(F^{c\,(a+b\,x)}\right)^n\right] dx$ when $m>0$

Derivation: Integration by parts

Basis: $\operatorname{Log}\left[1+e\,\left(F^{c\,(a+b\,x)}\right)^n\right] == -\partial_x \frac{\operatorname{PolyLog}\left[2,-e\,\left(F^{c\,(a+b\,x)}\right)^n\right]}{b\,c\,n\,\operatorname{Log}[F]}$

Rule: If $m>0$, then

$$\int (f+g\,x)^m \operatorname{Log}\left[1+e\,\left(F^{c\,(a+b\,x)}\right)^n\right] dx \longrightarrow -\frac{(f+g\,x)^m \operatorname{PolyLog}\left[2,\,-e\,\left(F^{c\,(a+b\,x)}\right)^n\right]}{b\,c\,n\,\operatorname{Log}[F]} + \frac{g\,m}{b\,c\,n\,\operatorname{Log}[F]} \int (f+g\,x)^{m-1} \operatorname{PolyLog}\left[2,\,-e\,\left(F^{c\,(a+b\,x)}\right)^n\right] dx$$

Program code:

```
Int[(f_.+g_.*x_)^m_.*Log[1+e_.*(F_^(c_.*(a_.+b_.*x_)))^n_.],x_Symbol] :=
  -(f+g*x)^m*PolyLog[2,-e*(F^(c*(a+b*x)))^n]/(b*c*n*Log[F]) +
  g*m/(b*c*n*Log[F])*Int[(f+g*x)^(m-1)*PolyLog[2,-e*(F^(c*(a+b*x)))^n],x] /;
FreeQ[{F,a,b,c,e,f,g,n},x] && GtQ[m,0]
```

2: $\int (f+g\,x)^m \operatorname{Log}\left[d+e\,\left(F^{c\,(a+b\,x)}\right)^n\right] dx$ when $m>0 \bigwedge d\neq 1$

Derivation: Integration by parts

Basis: $\partial_x \operatorname{Log}[d+e\,g[x]] == \partial_x \operatorname{Log}\left[1+\frac{e}{d}\,g[x]\right]$

Rule: If $m>0 \bigwedge d\neq 1$, then

$$\int (f+g\,x)^m \operatorname{Log}\left[d+e\,\left(F^{c\,(a+b\,x)}\right)^n\right] dx \longrightarrow \frac{(f+g\,x)^{m+1} \operatorname{Log}\left[d+e\,\left(F^{c\,(a+b\,x)}\right)^n\right]}{g\,(m+1)} - \frac{(f+g\,x)^{m+1} \operatorname{Log}\left[1+\frac{e}{d}\,\left(F^{c\,(a+b\,x)}\right)^n\right]}{g\,(m+1)} + \int (f+g\,x)^m \operatorname{Log}\left[1+\frac{e}{d}\,\left(F^{c\,(a+b\,x)}\right)^n\right] dx$$

Program code:

```
Int[(f_.+g_.*x_)^m_.*Log[d_+e_.*(F_^(c_.*(a_.+b_.*x_)))^n_.],x_Symbol] :=
  (f+g*x)^(m+1)*Log[d+e*(F^(c*(a+b*x)))^n]/(g*(m+1)) -
  (f+g*x)^(m+1)*Log[1+e/d*(F^(c*(a+b*x)))^n]/(g*(m+1)) +
  Int[(f+g*x)^m*Log[1+e/d*(F^(c*(a+b*x)))^n],x] /;
FreeQ[{F,a,b,c,d,e,f,g,n},x] && GtQ[m,0] && NeQ[d,1]
```

6. $\int u \operatorname{Log}\left[d + e x + f \sqrt{a + b x + c x^2}\right] dx$ when $e^2 - c f^2 == 0$

1: $\int \operatorname{Log}\left[d + e x + f \sqrt{a + b x + c x^2}\right] dx$ when $e^2 - c f^2 == 0$

Derivation: Integration by parts and algebraic simplification

- **Rule: If** $e^2 - c f^2 == 0$ **, then** $\frac{b f + 2 c f x + 2 e \sqrt{a + b x + c x^2}}{f (a + b x + c x^2) + (d + e x) \sqrt{a + b x + c x^2}} == -\frac{f^2 (b^2 - 4 a c)}{(2 d e - b f^2) (a + b x + c x^2) - f (b d - 2 a e + (2 c d - b e) x) \sqrt{a + b x + c x^2}}.$

- **Rule: If** $e^2 - c f^2 == 0$ **, then**

$$\int \operatorname{Log}\left[d + e x + f \sqrt{a + b x + c x^2}\right] dx \longrightarrow x \operatorname{Log}\left[d + e x + f \sqrt{a + b x + c x^2}\right] - \frac{1}{2} \int \frac{x \left(b f + 2 c f x + 2 e \sqrt{a + b x + c x^2}\right)}{f \left(a + b x + c x^2\right) + (d + e x) \sqrt{a + b x + c x^2}} dx$$

$$\longrightarrow x \operatorname{Log}\left[d + e x + f \sqrt{a + b x + c x^2}\right] + \frac{f^2 \left(b^2 - 4 a c\right)}{2} \int \frac{x}{\left(2 d e - b f^2\right) \left(a + b x + c x^2\right) - f (b d - 2 a e + (2 c d - b e) x) \sqrt{a + b x + c x^2}} dx$$

- **Program code:**

```
Int[Log[d_.+e_.*x_+f_.*Sqrt[a_.+b_.*x_+c_.*x_^2]],x_Symbol] :=
  x*Log[d+e*x+f*Sqrt[a+b*x+c*x^2]] +
  f^2*(b^2-4*a*c)/2*Int[x/((2*d*e-b*f^2)*(a+b*x+c*x^2)-f*(b*d-2*a*e+(2*c*d-b*e)*x)*Sqrt[a+b*x+c*x^2]),x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[e^2-c*f^2,0]
```

```
Int[Log[d_.+e_.*x_+f_.*Sqrt[a_.+c_.*x_^2]],x_Symbol] :=
  x*Log[d+e*x+f*Sqrt[a+c*x^2]] -
  a*c*f^2*Int[x/(d*e*(a+c*x^2)+f*(a*e-c*d*x)*Sqrt[a+c*x^2]),x] /;
FreeQ[{a,c,d,e,f},x] && EqQ[e^2-c*f^2,0]
```

2: $\int (g x)^m \operatorname{Log}\left[d + e x + f \sqrt{a + b x + c x^2}\right] dx$ when $e^2 - c f^2 == 0 \wedge m \neq -1$

Derivation: Integration by parts and algebraic simplification

- **Rule: If** $e^2 - c f^2 == 0$ **, then** $\frac{b f + 2 c f x + 2 e \sqrt{a + b x + c x^2}}{f (a + b x + c x^2) + (d + e x) \sqrt{a + b x + c x^2}} == -\frac{f^2 (b^2 - 4 a c)}{(2 d e - b f^2) (a + b x + c x^2) - f (b d - 2 a e + (2 c d - b e) x) \sqrt{a + b x + c x^2}}.$

- **Rule: If** $e^2 - c f^2 == 0 \wedge m \neq -1$**, then**

$$\int (g\,x)^m\,\mathrm{Log}\left[d+e\,x+f\,\sqrt{a+b\,x+c\,x^2}\,\right]dx \longrightarrow \frac{(g\,x)^{m+1}\,\mathrm{Log}\left[d+e\,x+f\,\sqrt{a+b\,x+c\,x^2}\,\right]}{g\,(m+1)}-\frac{1}{2\,g\,(m+1)}\int \frac{(g\,x)^{m+1}\left(b\,f+2\,c\,f\,x+2\,e\,\sqrt{a+b\,x+c\,x^2}\right)}{f\left(a+b\,x+c\,x^2\right)+(d+e\,x)\,\sqrt{a+b\,x+c\,x^2}}\,dx$$

$$\longrightarrow \frac{(g\,x)^{m+1}\,\mathrm{Log}\left[d+e\,x+f\,\sqrt{a+b\,x+c\,x^2}\,\right]}{g\,(m+1)}+\frac{f^2\left(b^2-4\,a\,c\right)}{2\,g\,(m+1)}\int \frac{(g\,x)^{m+1}}{\left(2\,d\,e-b\,f^2\right)\left(a+b\,x+c\,x^2\right)-f\,(b\,d-2\,a\,e+(2\,c\,d-b\,e)\,x)\,\sqrt{a+b\,x+c\,x^2}}\,dx$$

Program code:

```
Int[(g_.*x_)^m_.*Log[d_.+e_.*x_+f_.*Sqrt[a_.+b_.*x_+c_.*x_^2]],x_Symbol] :=
  (g*x)^(m+1)*Log[d+e*x+f*Sqrt[a+b*x+c*x^2]]/(g*(m+1)) +
  f^2*(b^2-4*a*c)/(2*g*(m+1))*Int[(g*x)^(m+1)/((2*d*e-b*f^2)*(a+b*x+c*x^2)-f*(b*d-2*a*e+(2*c*d-b*e)*x)*Sqrt[a+b*x+c*x^2]),x] /;
FreeQ[{a,b,c,d,e,f,g,m},x] && EqQ[e^2-c*f^2,0] && NeQ[m,-1] && IntegerQ[2*m]
```

```
Int[(g_.*x_)^m_.*Log[d_.+e_.*x_+f_.*Sqrt[a_.+c_.*x_^2]],x_Symbol] :=
  (g*x)^(m+1)*Log[d+e*x+f*Sqrt[a+c*x^2]]/(g*(m+1)) -
  a*c*f^2/(g*(m+1))*Int[(g*x)^(m+1)/(d*e*(a+c*x^2)+f*(a*e-c*d*x)*Sqrt[a+c*x^2]),x] /;
FreeQ[{a,c,d,e,f,g,m},x] && EqQ[e^2-c*f^2,0] && NeQ[m,-1] && IntegerQ[2*m]
```

```
Int[v_.*Log[d_.+e_.*x_+f_.*Sqrt[u_]],x_Symbol] :=
  Int[v*Log[d+e*x+f*Sqrt[ExpandToSum[u,x]]],x] /;
FreeQ[{d,e,f},x] && QuadraticQ[u,x] && Not[QuadraticMatchQ[u,x]] && (EqQ[v,1] || MatchQ[v,(g_.*x)^m_. /; FreeQ[{g,m},x]])
```

7. $\int \frac{\text{Log}[c\,x^n]^r\,(a\,x^m + b\,\text{Log}[c\,x^n]^q)^p}{x}\,dx$ when $r == q-1$

1: $\int \frac{\text{Log}[c\,x^n]^r}{x\,(a\,x^m + b\,\text{Log}[c\,x^n]^q)}\,dx$ when $r == q-1$

Derivation: Algebraic expansion and reciprocal rule for integration

Basis: $\int \frac{F'[x]+G'[x]}{F[x]+G[x]}\,dx == \text{Log}[F[x]+G[x]]$

Rule: If $r == q-1$**, then**

$$\int \frac{\text{Log}[c\,x^n]^r}{x\,(a\,x^m + b\,\text{Log}[c\,x^n]^q)}\,dx \rightarrow \frac{1}{b\,n\,q}\int \frac{a\,m\,x^m + b\,n\,q\,\text{Log}[c\,x^n]^r}{x\,(a\,x^m + b\,\text{Log}[c\,x^n]^q)}\,dx - \frac{a\,m}{b\,n\,q}\int \frac{x^{m-1}}{a\,x^m + b\,\text{Log}[c\,x^n]^q}\,dx$$

$$\rightarrow \frac{\text{Log}[a\,x^m + b\,\text{Log}[c\,x^n]^q]}{b\,n\,q} - \frac{a\,m}{b\,n\,q}\int \frac{x^{m-1}}{a\,x^m + b\,\text{Log}[c\,x^n]^q}\,dx$$

Program code:

```
Int[Log[c_.*x_^n_.]^r_./(x_*(a_.*x_^m_.+b_.*Log[c_.*x_^n_.]^q_)),x_Symbol] :=
  Log[a*x^m+b*Log[c*x^n]^q]/(b*n*q) - a*m/(b*n*q)*Int[x^(m-1)/(a*x^m+b*Log[c*x^n]^q),x] /;
FreeQ[{a,b,c,m,n,q,r},x] && EqQ[r,q-1]
```

2: $\int \frac{\text{Log}[c\,x^n]^r\,(a\,x^m + b\,\text{Log}[c\,x^n]^q)^p}{x}\,dx$ when $r == q-1 \wedge p \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $r == q-1 \wedge p \in \mathbb{Z}^+$**, then**

$$\int \frac{\text{Log}[c\,x^n]^r\,(a\,x^m + b\,\text{Log}[c\,x^n]^q)^p}{x}\,dx \rightarrow \int \frac{\text{Log}[c\,x^n]^r}{x}\,\text{ExpandIntegrand}[(a\,x^m + b\,\text{Log}[c\,x^n]^q)^p,\,x]\,dx$$

Program code:

```
Int[Log[c_.*x_^n_.]^r_.*(a_.*x_^m_.+b_.*Log[c_.*x_^n_.]^q_)^p_./x_,x_Symbol] :=
  Int[ExpandIntegrand[Log[c*x^n]^r/x,(a*x^m+b*Log[c*x^n]^q)^p,x],x] /;
FreeQ[{a,b,c,m,n,p,q,r},x] && EqQ[r,q-1] && IGtQ[p,0]
```

3: $\int \frac{\text{Log}[c\,x^n]^r\,(a\,x^m+b\,\text{Log}[c\,x^n]^q)^p}{x}\,dx$ when $r == q-1\,\land\,p \neq -1$

Derivation: Algebraic expansion and reciprocal rule for integration

Basis: $\int (F[x]+G[x])^p\,(F'[x]+G'[x])\,dx == \frac{(F[x]+G[x])^{p+1}}{p+1}$

Rule: If $r == q-1\,\land\,p \neq -1$**, then**

$$\int \frac{\text{Log}[c\,x^n]^r\,(a\,x^m+b\,\text{Log}[c\,x^n]^q)^p}{x}\,dx \longrightarrow$$

$$\frac{1}{b\,n\,q}\int \frac{(a\,m\,x^m+b\,n\,q\,\text{Log}[c\,x^n]^r)\,(a\,x^m+b\,\text{Log}[c\,x^n]^q)^p}{x}\,dx-\frac{a\,m}{b\,n\,q}\int x^{m-1}\,(a\,x^m+b\,\text{Log}[c\,x^n]^q)^p\,dx$$

$$\longrightarrow \frac{(a\,x^m+b\,\text{Log}[c\,x^n]^q)^{p+1}}{b\,n\,q\,(p+1)}-\frac{a\,m}{b\,n\,q}\int x^{m-1}\,(a\,x^m+b\,\text{Log}[c\,x^n]^q)^p\,dx$$

Program code:

```
Int[Log[c_.*x_^n_.]^r_.*(a_.*x_^m_.+b_.*Log[c_.*x_^n_.]^q_)^p_./x_,x_Symbol] :=
  (a*x^m+b*Log[c*x^n]^q)^(p+1)/(b*n*q*(p+1)) -
  a*m/(b*n*q)*Int[x^(m-1)*(a*x^m+b*Log[c*x^n]^q)^p,x] /;
FreeQ[{a,b,c,m,n,p,q,r},x] && EqQ[r,q-1] && NeQ[p,-1]
```

8. $\int \frac{(d\,x^m + e\,\text{Log}[c\,x^n]^r)\,(a\,x^m + b\,\text{Log}[c\,x^n]^q)^p}{x}\,dx$ when $r == q - 1$

1. $\int \frac{d\,x^m + e\,\text{Log}[c\,x^n]^r}{x\,(a\,x^m + b\,\text{Log}[c\,x^n]^q)}\,dx$ when $r == q - 1$

1: $\int \frac{d\,x^m + e\,\text{Log}[c\,x^n]^r}{x\,(a\,x^m + b\,\text{Log}[c\,x^n]^q)}\,dx$ when $r == q - 1 \land a\,e\,m - b\,d\,n\,q == 0$

Derivation: Reciprocal rule for integration

Basis: $\int \frac{F'[x]+G'[x]}{F[x]+G[x]}\,dx == \text{Log}[F[x] + G[x]]$

Rule: If $r == q - 1 \land a\,e\,m - b\,d\,n\,q == 0$**, then**

$$\int \frac{d\,x^m + e\,\text{Log}[c\,x^n]^r}{x\,(a\,x^m + b\,\text{Log}[c\,x^n]^q)}\,dx \longrightarrow \frac{e\,\text{Log}[a\,x^m + b\,\text{Log}[c\,x^n]^q]}{b\,n\,q}$$

Program code:

```
Int[(d_.*x_^m_.+e_.*Log[c_.*x_^n_.]^r_.)/(x_*(a_.*x_^m_.+b_.*Log[c_.*x_^n_.]^q_)),x_Symbol] :=
  e*Log[a*x^m+b*Log[c*x^n]^q]/(b*n*q) /;
FreeQ[{a,b,c,d,e,m,n,q,r},x] && EqQ[r,q-1] && EqQ[a*e*m-b*d*n*q,0]
```

```
Int[(u_+d_.*x_^m_.+e_.*Log[c_.*x_^n_.]^r_.)/(x_*(a_.*x_^m_.+b_.*Log[c_.*x_^n_.]^q_)),x_Symbol] :=
  e*Log[a*x^m+b*Log[c*x^n]^q]/(b*n*q) + Int[u/(x*(a*x^m+b*Log[c*x^n]^q)),x] /;
FreeQ[{a,b,c,d,e,m,n,q,r},x] && EqQ[r,q-1] && EqQ[a*e*m-b*d*n*q,0]
```

2: $\int \frac{d\,x^m + e\,\text{Log}[c\,x^n]^r}{x\,(a\,x^m + b\,\text{Log}[c\,x^n]^q)}\,dx$ when $r == q-1 \land a\,e\,m - b\,d\,n\,q \neq 0$

Derivation: Algebraic expansion and reciprocal rule for integration

Basis: $\int \frac{F'[x]+G'[x]}{F[x]+G[x]}\,dx == \text{Log}[F[x]+G[x]]$

Rule: If $r == q-1 \land a\,e\,m - b\,d\,n\,q \neq 0$**, then**

$$\int \frac{d\,x^m + e\,\text{Log}[c\,x^n]^r}{x\,(a\,x^m + b\,\text{Log}[c\,x^n]^q)}\,dx \longrightarrow \frac{e}{b\,n\,q}\int \frac{a\,m\,x^m + b\,n\,q\,\text{Log}[c\,x^n]^r}{x\,(a\,x^m + b\,\text{Log}[c\,x^n]^q)}\,dx - \frac{(a\,e\,m - b\,d\,n\,q)}{b\,n\,q}\int \frac{x^{m-1}}{a\,x^m + b\,\text{Log}[c\,x^n]^q}\,dx$$

$$\longrightarrow \frac{e\,\text{Log}[a\,x^m + b\,\text{Log}[c\,x^n]^q]}{b\,n\,q} - \frac{(a\,e\,m - b\,d\,n\,q)}{b\,n\,q}\int \frac{x^{m-1}}{a\,x^m + b\,\text{Log}[c\,x^n]^q}\,dx$$

Program code:

```
Int[(d_.*x_^m_.+e_.*Log[c_.*x_^n_.]^r_.)/(x_*(a_.*x_^m_.+b_.*Log[c_.*x_^n_.]^q_)),x_Symbol] :=
  e*Log[a*x^m+b*Log[c*x^n]^q]/(b*n*q) -
  (a*e*m-b*d*n*q)/(b*n*q)*Int[x^(m-1)/(a*x^m+b*Log[c*x^n]^q),x] /;
FreeQ[{a,b,c,d,e,m,n,q,r},x] && EqQ[r,q-1] && NeQ[a*e*m-b*d*n*q,0]
```

2. $\int \frac{(d\,x^m + e\,\text{Log}[c\,x^n]^r)\,(a\,x^m + b\,\text{Log}[c\,x^n]^q)^p}{x}\,dx$ when $r == q-1 \land p \neq -1$

1: $\int \frac{(d\,x^m + e\,\text{Log}[c\,x^n]^r)\,(a\,x^m + b\,\text{Log}[c\,x^n]^q)^p}{x}\,dx$ when $r == q-1 \land p \neq -1 \land a\,e\,m - b\,d\,n\,q == 0$

Derivation: Algebraic expansion and reciprocal rule for integration

Basis: $\int (F[x]+G[x])^p\,(F'[x]+G'[x])\,dx == \frac{(F[x]+G[x])^{p+1}}{p+1}$

Rule: If $r == q-1 \land p \neq -1 \land a\,e\,m - b\,d\,n\,q == 0$**, then**

$$\int \frac{(d\,x^m + e\,\text{Log}[c\,x^n]^r)\,(a\,x^m + b\,\text{Log}[c\,x^n]^q)^p}{x}\,dx \longrightarrow \frac{e\,(a\,x^m + b\,\text{Log}[c\,x^n]^q)^{p+1}}{b\,n\,q\,(p+1)}$$

Program code:

```
Int[(d_.*x_^m_.+e_.*Log[c_.*x_^n_.]^r_.)*(a_.*x_^m_.+b_.*Log[c_.*x_^n_.]^q_)^p_./x_,x_Symbol] :=
  e*(a*x^m+b*Log[c*x^n]^q)^(p+1)/(b*n*q*(p+1)) /;
FreeQ[{a,b,c,d,e,m,n,p,q,r},x] && EqQ[r,q-1] && NeQ[p,-1] && EqQ[a*e*m-b*d*n*q,0]
```

2: $\int \frac{(d\,x^m + e\,\text{Log}[c\,x^n]^r)\,(a\,x^m + b\,\text{Log}[c\,x^n]^q)^p}{x}\,dx$ when $r == q-1 \land p \neq -1 \land a\,e\,m - b\,d\,n\,q \neq 0$

Derivation: Algebraic expansion and reciprocal rule for integration

Basis: $\int (F[x] + G[x])^p\,(F'[x] + G'[x])\,dx == \frac{(F[x]+G[x])^{p+1}}{p+1}$

Rule: If $r == q-1 \land p \neq -1 \land a\,e\,m - b\,d\,n\,q \neq 0$**, then**

$$\int \frac{(d\,x^m + e\,\text{Log}[c\,x^n]^r)\,(a\,x^m + b\,\text{Log}[c\,x^n]^q)^p}{x}\,dx \longrightarrow$$

$$\frac{e}{b\,n\,q}\int \frac{(a\,m\,x^m + b\,n\,q\,\text{Log}[c\,x^n]^r)\,(a\,x^m + b\,\text{Log}[c\,x^n]^q)^p}{x}\,dx - \frac{(a\,e\,m - b\,d\,n\,q)}{b\,n\,q}\int x^{m-1}\,(a\,x^m + b\,\text{Log}[c\,x^n]^q)^p\,dx$$

$$\longrightarrow \frac{e\,(a\,x^m + b\,\text{Log}[c\,x^n]^q)^{p+1}}{b\,n\,q\,(p+1)} - \frac{(a\,e\,m - b\,d\,n\,q)}{b\,n\,q}\int x^{m-1}\,(a\,x^m + b\,\text{Log}[c\,x^n]^q)^p\,dx$$

Program code:

```
Int[(d_.*x_^m_.+e_.*Log[c_.*x_^n_.]^r_.)*(a_.*x_^m_.+b_.*Log[c_.*x_^n_.]^q_)^p_./x_,x_Symbol] :=
  e*(a*x^m+b*Log[c*x^n]^q)^(p+1)/(b*n*q*(p+1)) -
  (a*e*m-b*d*n*q)/(b*n*q)*Int[x^(m-1)*(a*x^m+b*Log[c*x^n]^q)^p,x] /;
FreeQ[{a,b,c,d,e,m,n,p,q,r},x] && EqQ[r,q-1] && NeQ[p,-1] && NeQ[a*e*m-b*d*n*q,0]
```

9: $\int \frac{d\,x^m + e\,x^m\,\text{Log}[c\,x^n] + f\,\text{Log}[c\,x^n]^q}{x\,(a\,x^m + b\,\text{Log}[c\,x^n]^q)^2}\,dx$ when $e\,n + d\,m == 0 \land a\,f + b\,d\,(q-1) == 0$

Rule: If $e\,n + d\,m == 0 \land a\,f + b\,d\,(q-1) == 0$**, then**

$$\int \frac{d\,x^m + e\,x^m\,\text{Log}[c\,x^n] + f\,\text{Log}[c\,x^n]^q}{x\,(a\,x^m + b\,\text{Log}[c\,x^n]^q)^2}\,dx \longrightarrow \frac{d\,\text{Log}[c\,x^n]}{a\,n\,(a\,x^m + b\,\text{Log}[c\,x^n]^q)}$$

Program code:

```
Int[(d_.*x_^m_.+e_.*x_^m_.*Log[c_.*x_^n_.]+f_.*Log[c_.*x_^n_.]^q_.)/(x_*(a_.*x_^m_.+b_.*Log[c_.*x_^n_.]^q_)^2),x_Symbol] :=
  d*Log[c*x^n]/(a*n*(a*x^m+b*Log[c*x^n]^q)) /;
FreeQ[{a,b,c,d,e,f,m,n,q},x] && EqQ[e*n+d*m,0] && EqQ[a*f+b*d*(q-1),0]
```

10: $\int \frac{d + e \operatorname{Log}[c x^n]}{(a x + b \operatorname{Log}[c x^n]^q)^2} \, dx$ when $d + e n q == 0$

Derivation: Algebraic expansion

Rule: If $d + e n q == 0$**, then**

$$\int \frac{d + e \operatorname{Log}[c x^n]}{(a x + b \operatorname{Log}[c x^n]^q)^2} \, dx \longrightarrow -\frac{1}{a} \int \frac{a e n x - a e x \operatorname{Log}[c x^n] + b (d + e n) \operatorname{Log}[c x^n]^q}{x (a x + b \operatorname{Log}[c x^n]^q)^2} \, dx + \frac{d + e n}{a} \int \frac{1}{x (a x + b \operatorname{Log}[c x^n]^q)} \, dx$$

$$\longrightarrow -\frac{e \operatorname{Log}[c x^n]}{a (a x + b \operatorname{Log}[c x^n]^q)} + \frac{d + e n}{a} \int \frac{1}{x (a x + b \operatorname{Log}[c x^n]^q)} \, dx$$

Program code:

```
Int[(d_+e_.*Log[c_.*x_^n_.])/(a_.*x_+b_.*Log[c_.*x_^n_.]^q_)^2,x_Symbol] :=
  -e*Log[c*x^n]/(a*(a*x+b*Log[c*x^n]^q)) + (d+e*n)/a*Int[1/(x*(a*x+b*Log[c*x^n]^q)),x] /;
FreeQ[{a,b,c,d,e,n,q},x] && EqQ[d+e*n*q,0]
```

11. $\int v \operatorname{Log}[u] \, dx$ when u is free of inverse functions

1: $\int \operatorname{Log}[u] \, dx$ when u is free of inverse functions

Reference: A&S 4.1.53

Derivation: Integration by parts

Rule: If `InverseFunctionFreeQ[u, x]`**, then**

$$\int \operatorname{Log}[u] \, dx \longrightarrow x \operatorname{Log}[u] - \int \frac{x \, \partial_x u}{u} \, dx$$

Program code:

```
Int[Log[u_],x_Symbol] :=
  x*Log[u] - Int[SimplifyIntegrand[x*D[u,x]/u,x],x] /;
InverseFunctionFreeQ[u,x]
```

```
Int[Log[u_],x_Symbol] :=
  x*Log[u] - Int[SimplifyIntegrand[x*Simplify[D[u,x]/u],x],x] /;
ProductQ[u]
```

2. $\int (a+b\,x)^m \,\text{Log}[u]\, dx$ when **u is free of inverse functions**

1: $\int \frac{\text{Log}[u]}{a+b\,x}\, dx$ when $\text{RationalFunctionQ}\left[\frac{\partial_x u}{u}, x\right]$

Reference: G&R 2.727.2

Derivation: Integration by parts

Basis: $\frac{1}{a+b\,x} == \partial_x \frac{\text{Log}[a+b\,x]}{b}$

Rule: If $\text{RationalFunctionQ}\left[\frac{\partial_x u}{u}, x\right]$**, then**

$$\int \frac{\text{Log}[u]}{a+b\,x}\, dx \longrightarrow \frac{\text{Log}[a+b\,x]\,\text{Log}[u]}{b} - \frac{1}{b}\int \frac{\text{Log}[a+b\,x]\,\partial_x u}{u}\, dx$$

Program code:

```
Int[Log[u_]/(a_.+b_.*x_),x_Symbol] :=
  Log[a+b*x]*Log[u]/b -
  1/b*Int[SimplifyIntegrand[Log[a+b*x]*D[u,x]/u,x],x] /;
FreeQ[{a,b},x] && RationalFunctionQ[D[u,x]/u,x] && (NeQ[a,0] || Not[BinomialQ[u,x] && EqQ[BinomialDegree[u,x]^2,1]])
```

2: $\int (a+b\,x)^m \,\text{Log}[u]\, dx$ when $\text{InverseFunctionFreeQ}[u, x] \land m \neq -1$

Reference: G&R 2.725.1, A&S 4.1.54

Derivation: Integration by parts

Basis: $(a+b\,x)^m == \partial_x \frac{(a+b\,x)^{m+1}}{b\,(m+1)}$

Rule: If $\text{InverseFunctionFreeQ}[u, x] \land m \neq -1$**, then**

$$\int (a+b\,x)^m \,\text{Log}[u]\, dx \longrightarrow \frac{(a+b\,x)^{m+1}\,\text{Log}[u]}{b\,(m+1)} - \frac{1}{b\,(m+1)}\int \frac{(a+b\,x)^{m+1}\,\partial_x u}{u}\, dx$$

Program code:

```
Int[(a_.+b_.*x_)^m_.*Log[u_],x_Symbol] :=
  (a+b*x)^(m+1)*Log[u]/(b*(m+1)) -
  1/(b*(m+1))*Int[SimplifyIntegrand[(a+b*x)^(m+1)*D[u,x]/u,x],x] /;
FreeQ[{a,b,m},x] && InverseFunctionFreeQ[u,x] && NeQ[m,-1] (* && Not[FunctionOfQ[x^(m+1),u,x]] && FalseQ[PowerVariableExpn[u,m+1,x]
```

3: $\int \frac{\text{Log}[u]}{Q_x} \, dx$ when QuadraticQ[Q_x] $\wedge$ InverseFunctionFreeQ[u, x]

Derivation: Integration by parts

Rule: If QuadraticQ[Q_x] $\wedge$ InverseFunctionFreeQ[u, x], let $v = \int \frac{1}{Q_x} \, dx$, then

$$\int \frac{\text{Log}[u]}{Q_x} \, dx \longrightarrow v \, \text{Log}[u] - \int \frac{v \, \partial_x u}{u} \, dx$$

Program code:

```
Int[Log[u_]/Qx_,x_Symbol] :=
  With[{v=IntHide[1/Qx,x]},
  v*Log[u] - Int[SimplifyIntegrand[v*D[u,x]/u,x],x]] /;
QuadraticQ[Qx,x] && InverseFunctionFreeQ[u,x]
```

4: $\int u^{a\,x} \, \text{Log}[u] \, dx$ when u is free of inverse functions

Basis: $u^{a\,x} \, \text{Log}[u] = \frac{\partial_x u^{a\,x}}{a} - x \, u^{a\,x-1} \, \partial_x u$

Rule: If InverseFunctionFreeQ[u, x], then

$$\int u^{a\,x} \, \text{Log}[u] \, dx \longrightarrow \frac{u^{a\,x}}{a} - \int x \, u^{a\,x-1} \, \partial_x u \, dx$$

Program code:

```
Int[u_^(a_.*x_)*Log[u_],x_Symbol] :=
  u^(a*x)/a - Int[SimplifyIntegrand[x*u^(a*x-1)*D[u,x],x],x] /;
FreeQ[a,x] && InverseFunctionFreeQ[u,x]
```

5: $\int v\,\mathrm{Log}[u]\,dx$ when u and $\int v\,dx$ are free of inverse functions

Derivation: Integration by parts

Rule: If InverseFunctionFreeQ[u, x], **let** $w = \int v\,dx$, **if** InverseFunctionFreeQ[w, x], **then**

$$\int v\,\mathrm{Log}[u]\,dx \longrightarrow w\,\mathrm{Log}[u] - \frac{1}{b}\int \frac{w\,\partial_x u}{u}\,dx$$

Program code:

```
Int[v_*Log[u_],x_Symbol] :=
  With[{w=IntHide[v,x]},
  Dist[Log[u],w,x] - Int[SimplifyIntegrand[w*D[u,x]/u,x],x] /;
 InverseFunctionFreeQ[w,x]] /;
InverseFunctionFreeQ[u,x]
```

```
Int[v_*Log[u_],x_Symbol] :=
  With[{w=IntHide[v,x]},
  Dist[Log[u],w,x] - Int[SimplifyIntegrand[w*Simplify[D[u,x]/u],x],x] /;
 InverseFunctionFreeQ[w,x]] /;
ProductQ[u]
```

12. $\int u\,\mathrm{Log}[v]\,\mathrm{Log}[w]\,dx$ when v, w and $\int u\,dx$ are free of inverse functions

1: $\int \mathrm{Log}[v]\,\mathrm{Log}[w]\,dx$ when v and w are free of inverse functions

Derivation: Integration by parts

Rule: If v and w are free of inverse functions, then

$$\int \mathrm{Log}[v]\,\mathrm{Log}[w]\,dx \longrightarrow x\,\mathrm{Log}[v]\,\mathrm{Log}[w] - \int \frac{x\,\mathrm{Log}[w]\,\partial_x v}{v}\,dx - \int \frac{x\,\mathrm{Log}[v]\,\partial_x w}{w}\,dx$$

Program code:

```
Int[Log[v_]*Log[w_],x_Symbol] :=
  x*Log[v]*Log[w] -
  Int[SimplifyIntegrand[x*Log[w]*D[v,x]/v,x],x] -
  Int[SimplifyIntegrand[x*Log[v]*D[w,x]/w,x],x] /;
InverseFunctionFreeQ[v,x] && InverseFunctionFreeQ[w,x]
```

2: $\int u \operatorname{Log}[v] \operatorname{Log}[w] \, dx$ when v, w and $\int u \, dx$ are free of inverse functions

Derivation: Integration by parts

Rule: If v and w are free of inverse functions, let $z == \int u \, dx$, if z is free of inverse functions, then

$$\int u \operatorname{Log}[v] \operatorname{Log}[w] \, dx \longrightarrow z \operatorname{Log}[v] \operatorname{Log}[w] - \int \frac{z \operatorname{Log}[w] \, \partial_x v}{v} \, dx - \int \frac{z \operatorname{Log}[v] \, \partial_x w}{w} \, dx$$

Program code:

```
Int[u_*Log[v_]*Log[w_],x_Symbol] :=
  With[{z=IntHide[u,x]},
  Dist[Log[v]*Log[w],z,x] -
  Int[SimplifyIntegrand[z*Log[w]*D[v,x]/v,x],x] -
  Int[SimplifyIntegrand[z*Log[v]*D[w,x]/w,x],x] /;
 InverseFunctionFreeQ[z,x]] /;
InverseFunctionFreeQ[v,x] && InverseFunctionFreeQ[w,x]
```

13: $\int f^{a \operatorname{Log}[u]} \, dx$

Derivation: Algebraic simplification

Basis: $f^{a \operatorname{Log}[g]} = g^{a \operatorname{Log}[f]}$

Rule:

$$\int f^{a \operatorname{Log}[u]} \, dx \longrightarrow \int u^{a \operatorname{Log}[f]} \, dx$$

Program code:

```
Int[f_^(a_.*Log[u_]),x_Symbol] :=
  Int[u^(a*Log[f]),x] /;
FreeQ[{a,f},x]
```

14: $\int \frac{F[\text{Log}[a\,x^n]]}{x}\,dx$

Derivation: Integration by substitution

- **Basis:** $\frac{F[\text{Log}[a\,x^n]]}{x} = \frac{1}{n}\,F[\text{Log}[a\,x^n]]\,\partial_x \text{Log}[a\,x^n]$

Rule:

$$\int \frac{F[\text{Log}[a\,x^n]]}{x}\,dx \longrightarrow \frac{1}{n}\,\text{Subst}\left[\int F[x]\,dx,\,x,\,\text{Log}[a\,x^n]\right]$$

Program code:

```
(* If[TrueQ[$LoadShowSteps],

Int[u_/x_,x_Symbol] :=
  With[{lst=FunctionOfLog[u,x]},
  ShowStep["","Int[F[Log[a*x^n]]/x,x]","Subst[Int[F[x],x],x,Log[a*x^n]]/n",Hold[
  1/lst[[3]]*Subst[Int[lst[[1]],x],x,Log[lst[[2]]]]]] /;
 Not[FalseQ[lst]]] /;
SimplifyFlag && NonsumQ[u],

Int[u_/x_,x_Symbol] :=
  With[{lst=FunctionOfLog[u,x]},
  1/lst[[3]]*Subst[Int[lst[[1]],x],x,Log[lst[[2]]]] /;
 Not[FalseQ[lst]]] /;
NonsumQ[u]] *)
```

```
If[TrueQ[$LoadShowSteps],

Int[u_,x_Symbol] :=
  With[{lst=FunctionOfLog[Cancel[x*u],x]},
  ShowStep["","Int[F[Log[a*x^n]]/x,x]","Subst[Int[F[x],x],x,Log[a*x^n]]/n",Hold[
  1/lst[[3]]*Subst[Int[lst[[1]],x],x,Log[lst[[2]]]]]] /;
 Not[FalseQ[lst]]] /;
SimplifyFlag && NonsumQ[u],

Int[u_,x_Symbol] :=
  With[{lst=FunctionOfLog[Cancel[x*u],x]},
  1/lst[[3]]*Subst[Int[lst[[1]],x],x,Log[lst[[2]]]] /;
 Not[FalseQ[lst]]] /;
NonsumQ[u]]
```

15: $\int \mathbf{u\ Log[Gamma[v]]}\, dx$

- **Derivation: Piecewise constant extraction**
- **Basis:** ∂_x (**Log[Gamma[F[x]]] – LogGamma[F[x]]**) == **0**
- **Rule:**

$$\int \text{u Log[Gamma[v]]}\, dx \longrightarrow (\text{Log[Gamma[v]]} - \text{LogGamma[v]}) \int u\, dx + \int \text{u LogGamma[v]}\, dx$$

- **Program code:**

```
Int[u_.*Log[Gamma[v_]],x_Symbol] :=
  (Log[Gamma[v]]-LogGamma[v])*Int[u,x] + Int[u*LogGamma[v],x]
```

N: $\int u\,(a\,x^m + b\,x^r\,\text{Log}[c\,x^n]^q)^p\,dx$ when $p \in \mathbb{Z}$

- **Derivation: Algebraic normalization**
- **Rule: If $p \in \mathbb{Z}$, then**

$$\int u\,(a\,x^m + b\,x^r\,\text{Log}[c\,x^n]^q)^p\,dx \longrightarrow \int u\,x^{p\,r}\,(a\,x^{m-r} + b\,\text{Log}[c\,x^n]^q)^p\,dx$$

- **Program code:**

```
Int[u_.*(a_.*x_^m_.+b_.*x_^r_.*Log[c_.*x_^n_.]^q_.)^p_.,x_Symbol] :=
  Int[u*x^(p*r)*(a*x^(m-r)+b*Log[c*x^n]^q)^p,x] /;
FreeQ[{a,b,c,m,n,p,q,r},x] && IntegerQ[p]
```